

4.5 特殊関数とライブラリ CML[要 full version]

ver. 0.3

2回にわたって説明しましたMAXLIKは、制約条件なしにLog-Likelihood関数を最大化するような係数を求めることに特化したライブラリでした。それに対して、各係数パラメータに制約条件があるような条件のもとで、Log-Likelihood関数を最大化するような係数を求めるのがCMLライブラリです。名前のとおり、Constrained Maximum Likelihoodを求めるものです。もちろん、これには制約条件なしの計算も可能なので、MAXLIKで本来は行なう計算もCMLで行なわれていることも多々あります。文法は、ほとんどMAXLIKと同じで、グローバル変数の名前が_max何々となっていたところが、今度は_cml何々と変更になります。また、係数パラメータに制約を加える場合には、sqpSolveでやったような等号制約、不等号制約、それにバウンダリー条件を行列の形で表現して加えます。

まずは、前回MAXLIKで行なった計算をCMLで行なってみましょう。sのところに正值条件を加えてみます。Dドライブにdatafile1.txtというyとxのシリーズデータがあるものとします。これをCMLライブラリを呼び出して、CMLの制限条件つき最大化で計算させてみます。

プログラム

```
new; cls;
library cml;
cmlset;
load data[29,2]=d:datafile1.txt;
data=data[2:29,.];

start={0,0,1};
_cml_Bounds={-1e256 1e256,
              -1e256 1e256,
              0.001 1e256};
_cml_Algorithm = 1;
_cml_ParNames = {"CONST","X","s"};
call cmlprt(cml(data,0,&ll,start));

proc ll(b,data);
  local beta, s, y, x, e;
  y=data[.,1];
  x=ones(28,1)~data[.,2];
  beta=b[1:2];
  s=b[3];
```

```

e=y-x*beta;
retp(-1/2*ln(2*pi)-1/2*ln(s^2)-1/2*e^2/s^2);
endp;

```

画面表示

```

return code =    0
normal convergence

```

```

Mean log-likelihood      -5.33947
Number of cases          28

```

Covariance of the parameters computed by the following method:
Inverse of computed Hessian

Parameters	Estimates	Std. err.	Gradient
CONST	77.7952	21.2357	0.0000
X	52.0098	3.6923	0.0000
s	50.4274	6.7387	-0.0000

```

Number of iterations      93
Minutes to convergence    0.17933

```

上のプログラムは、MAXLIKのものと同じものです。まず、library cml;でCMLライブラリを呼び出して、そこにある関数群を使用可能にします。そして、cmlset;でCMLで使われる諸設定に関するグローバル変数を初期化します。スタート値を設定した後、バウンダリー条件を設定します。これは、spqSolveの時と全く同じように、この場合、3つのパラメータがありますから、3行2列の_cml_Boundsの行列値を設定することになります。1行目は、第1パラメータについての最小値と最大値を、2行目は第2パラメータの最小値と最大値を、そして3行目は、第3パラメータ（すなわちs）の最小値と最大値を設定しています。通常、慣習として、下天井は-1e256を、上天井は1e256を使用します。常に下限と上限を設定しなければなりません。第1パラメータおよび第2パラメータは制約がないので、-1e256から1e256までの値を設定しています。第3パラメータsは0より大きい数なので、0からではなくて、0から有効数字程度分大きい、たとえば0.001などの数字から1e256までとしています。MAXLIKの際にアルゴリズム2であったBFGSは、CMLではアルゴリズム1となり、_cml_Algorithm = 1; と設定します。なお、CMLでは、アルゴ

リズム 1 はBFGS、2 はDEF、3 はNEWTON、4 はBHHHとなっています（ここでは、CMLバージョン1.0を用いています）。独立変数およびその他のパラメータの設定は、`_cml_ParNames`で行ないます。MAXLINKで`_max_ParNames`を設定したものを、`cml`とするだけの変更となります。そして、Log-Likelihood関数`ll`を最大化するために、`call cmlprt(cml(data,0,&ll,start));`とします。MAXLIKで、`maxlik`および`maxprt`となっていた関数名を、CMLでは`cml`および`cmlprt`に変更するだけです。結果は、若干アルゴリズムが違うため、係数の若干の違いは認められますが、MAXLIKのときとほぼ同じになっています。しかも、`s` が負の数の場合には`iteration`は回らなく、この設定の場合、`s` が0.001以上の場合についてのみ`iteration`が行なわれます。

```

ポイント  _cml_Bounds = { 第 1 パラメータの下限    第 1 パラメータの上限 ,
                           第 2 パラメータの下限    第 2 パラメータの上限 ,
                           .....
                           第 K パラメータの下限    第 K パラメータの上限  };

```

ただし、慣習上 - には-e256、+ には1e256を使う。

等号をとまなわない下限または上限には、たとえば、 0.0001 を $0 <$ または < 0 の意味で使う。小数の大きさは任意。

次に、前章で扱った標準TOBITモデルについて、上の計算と同様に、 s のところに正値の制約を加えた計算をしてみます。

プログラム

```
new; cls;
library cml;
cmlset;
load data[51,7]=d:datafile6.txt;
data=data[2:51,.];

hour=data[,2];
one=ones(50,1);
c18=data[,4];
age=data[,5];
age2=age^2;
ed1=data[,6];
hi=data[,7];
data=hour~one~c18~age~age2~ed1~hi;
```

```

_cml_Bounds = {-1e256  1e256,
               -1e256  1e256,
               -1e256  1e256,
               -1e256  1e256,
               -1e256  1e256,
               -1e256  1e256,
               0.0001  1e256};
_cml_ParNames = {"CONST","C18","AGE","AGE2","ED","HI","SIGMA"};
_cml_Algorithm = 4;
start={-1,0,0,0,0,0,1};
call cmlprt(cml(data,0,&ll,start));

proc ll(b,data);
    local y,x,s,beta;
    y=data[,1];
    x=data[,2:7];
    s=b[7];
    beta=b[1:6];
    retp( (y==0).*ln(1-cdfn(x*beta/s))+ (y.>0).*lnpdfmvn(y-x*beta,s^2) );
endp;

```

画面表示

return code = 0

normal convergence

Mean log-likelihood -3.16591

Number of cases 50

Covariance of the parameters computed by the following method:

Inverse of computed Hessian

Parameters	Estimates	Std. err.	Gradient
CONST	-47.4330	45.8176	-0.0000
C18	-7.1220	3.1963	-0.0000

AGE	2.8162	2.3676	-0.0000
AGE2	-0.0432	0.0273	-0.0000
ED	2.5322	1.3753	-0.0000
HI	0.0001	0.0002	-0.0000
SIGMA	22.4303	3.0444	0.0000

Number of iterations 1074
 Minutes to convergence 3.26883

MAXLIKからの変更点は、CMLライブラリを呼び出して、`cmlset;`でそのグローバル変数設定を初期化するほか、いままで`_max`何々となっていたグローバル変数を`_cml`何々に変更します。そして、この場合、`_cml_Bounds`というパラメータのバウンダリーを設定するところで、7番目のパラメータ、すなわち`s`を0.0001から上天井の`1e256`までとしています。その他のパラメータについては、`-1e256`から`1e256`までとして事実上無制約にします。アルゴリズムの番号は、MAXLIKとはずれていますから、BHHHの場合は`_cml_Algorithm = 4;`というように4に変更します。そして、`call cmlprt(cml(data,0,&ll,start));`という`cml`とその値を表示する`cmlprt`とが2重に入れ子にした関数を`call`で呼んでいます。ここでも結果はMAXLIKと若干iterationの数、係数ともに若干の違いはありますが、ほとんど同じ結果になっています。正值制限を加えてやると、ほかのアルゴリズムでも`s`がマイナスになって計算が続くということはありませんが、依然としてMAXLIK同様にアルゴリズムとスタートベクトルの組み合わせによっては収束しないという点は残ります。

ポイント CMLの書式

```
{x,f,g,cov,retcode} = cmlprt(cml(dataset,0,&ll,start));  
または、  
call cmlprt(cml(dataset,0,&ll,start));
```

また、MAXLIKと同じように、収束の許容桁数、最大iteration数、最大iteration分の時間数を次のようなグローバル値のデフォルト値を変更することによって設定できます。

```
_cml_DirTol = 1e-5;      ( 収束の許容桁数 )  
_cml_MaxIters = 1e+5;    ( 最大iteration数 )  
_cml_MaxTime = 1e+5;    ( 最大iterationに要する分数 )
```

ここで、`s`が0よりも大きいというバウンダリー条件だけではなくて、さらにいくつかの不等号、等号条件をもっているよく使われる例について説明します。

GARCH(1,1)

CMLの応用例としては、最も有力なのが、金融やマクロの時系列データにおいて残差項に一定の関係があって、そのパラメータに一定の条件関係があるGARCHファミリーのケースがあります。まずここで挙げるGARCH(p,q)は、

$$Y_t = X_t \beta + u_t$$

where $u_t = \sqrt{h_t} \cdot v_t$ ここで v_t は unit normal density

$$h_t = a_0 + b_1 h_{t-1} + b_2 h_{t-2} + \dots + b_p h_{t-p} \\ + a_1 u_{t-1}^2 + a_2 u_{t-2}^2 + \dots + a_q u_{t-q}^2$$

としたときに、次のような条件がパラメータにあるものをGARCHと呼びます。

$$a_0 > 0$$

$$b > 0$$

$$a > 0$$

$$\sum_{i=1}^p b_i + \sum_{j=1}^q a_j < 1$$

ここでは、 $p = 1$ 、 $q = 1$ の単純な例についてのプログラムをしてみます。この場合、およびは1個ずつで、バウンダリー条件のほかに、 $b + a < 1$ という不等号制約が加わります。今、Dドライブに、datafile14.txtというy、xの2シリーズの時系列データがそれぞれ100個の行だけラベルなしにあるものとします。

プログラム

```
new; cls;
```

```
library cml;
```

```
cmlset;
```

```
load data[100,2]=d:datafile14.txt;
```

```
data=data~ones(100,1);
```

```
_cml_Bounds={-1e256 1e256,
```

```
              -1e256 1e256,
```

```
              0.001 1e256,
```

```
              0          1,
```

```
              0          1};
```

```
_cml_c={0 0 0 -1 -1}; _cml_d= -0.9999;
```

```
_cml_ParNames = {"const","beta","a0","a1","b1"};
```

```
_cml_Algorithm=4;
```

```
b={0,0,1,1,1};
```

```
call cmlprt(cml(data,0,&ll,b));
```

```

proc ll(b,data);
  local y,x,e2,beta,a0,a1,b1,h;
  y=data[:,1]; x=data[:,3]~data[:,2]; beta=b[1:2];
  e2=(y-x*beta)^2;
  a0=b[3]; a1=b[4]; b1=b[5];
  h=recserrar(a0+lagn(e2,1)*a1, meanc(e2),b1);
  retp(-1/2*ln(2*pi)-1/2*ln(h)-1/2*e2./h);
endp;

```

画面表示

```

return code =    0
normal convergence

```

```

Mean log-likelihood      -0.482068
Number of cases         100

```

Covariance of the parameters computed by the following method:
Inverse of computed Hessian

Parameters	Estimates	Std. err.	Gradient
const	0.4772	0.0377	-0.0000
beta	1.9985	0.0130	-0.0000
a0	0.0611	0.0409	0.0000
a1	0.2089	0.1670	0.0000
b1	0.4134	0.2513	0.0000

```

Number of iterations    29
Minutes to convergence  0.06133

```

上のように、Log-Likelihood関数を作るのは、残差項の係数にARMAのようなプログラムを必要とするので容易ではありませんが、その部分をまず上のように与えられたものとして、そのLog-Likelihood関数を条件付の最大化をしてみます。bには、その1、2番目にはxの係数がきて、3番目以降、a0,a1,b1の順で設定されていて、それぞれに制約を課します。ここで、_cml_Bounds=のところは、第1、第2パラメータは無制約、第3パラメータのa0は0よりも大きいからです、下限は0ではなくて0.001にしています。第4、第5パラメータは0を含んでそれ以上ですから、ともに下限は0に設定されます。上限はGARCHの定常

性から足し合わせたものが1より小さくなりますから、それぞれも1が上限になります。
最適値問題の章のsqpSolveのところでやったのと同様に、不等号の制約は

$$_cml_C * b \geq _cml_D$$

の関係が成り立ちますから、符号の向きが逆の $_cml_C * b < _cml_D$ の場合には、両辺に - 1 をかけて、しかも等号を含まない不等号ですから、 - 1 ではなくて、-0.9999を採用しています。第3、第4パラメータにかかる係数はマイナスの符号をつけて-1と-1になっていて、その他の部分は0が入ります。すなわち、

$$\begin{bmatrix} 0 & 0 & 0 & -1 & -1 \end{bmatrix} \begin{bmatrix} const \\ beta \\ a0 \\ a1 \\ b1 \end{bmatrix} \geq -0.9999$$

とプログラムでは表していることになり、それは $-a1 - b1 \geq -0.9999$ のことであって、これは両辺に - 1 をかけると $a1 + b1 \leq 0.9999$ となって、近似的に $a1 + b1 < 1$ を表している。これらのパラメータ制約のもとで、アルゴリズム4のBHHHでLog-Likelihood関数を最大化するベクトルbをCMLで求めて、cmlprtで画面表示させています。

さて、Log-Likelihood関数のprocedureのところですが、これは基本的に2つの式の連立になっています。まず、 $h_t = a1 + b1 h_{t-1} + a1 u_{t-1}^2$ というパラメータの制約があります。これに対して、 v_t はunit normalなdesityですから、通常のLog-Likelihood関数の e^2 のところに、ルートの h_t を2乗して1をかけたもの、すなわち、 h_t 自身を代わりに代入します。

$$LL = -\frac{1}{2} \ln(2\pi) - \frac{1}{2} \ln(h) - \frac{e^2}{2h} \quad \text{where } e = y - x\beta$$

したがって、Log-Likelihoodのprocedureのリターンが $-1/2 * \ln(2 * \pi) - 1/2 * \ln(h) - 1/2 * e2./h$ というふうに足し合わせる前の列ベクトルになっています。ここまでは、容易にプログラムをたてられると思います。問題は、

$$h_t = a0 + b1 h_{t-1} + a1 u_{t-1}^2$$

の部分です。これはARプロセスを作る組込み関数recserarと、 v_t がunit normalなdesityであることを利用してe2の部分の列ベクトルを1つずらす方法で、 $e2(t-1)$ 、すなわちここでは、 u_{t-1}^2 のシリーズを作り出すという2つの合わせ技でhの式を書きます。まず、前半部分のp=1に相当する部分は元のe2の列ベクトルを1つずらします。これにはlagnの組込み関数を用います。たとえば、lagn(a,1)とすると、aという列ベクトルを下方方向に1だけずらし

て、ずれてなくなった先頭のところを0で埋め合わせる働きをします。そこで、aをたとえば1から20までの数列として、これを1つ下にずらして、一番最初に0で穴埋めすると、0から19までの列ベクトルの数列が生成されます。この1つずらす箇所を利用して、 $q = 1$ の部分をまず作ります。この1つずらしたシリーズにa1をかけて、さらにこれにa0を加えることによって、ARプロセスの組込み関数で動かない定数項とします。それが、ARプロセスを生成する組込み関数recserarの第1要素になります。そして、その第2要素をもとにしてAR項を作り、第3要素のb1がスケーラーの場合には $q = 1$ の場合の $b_1 h_{t-1}$ が生成されます。この場合の第2要素には、meanc(e2)がきています。これらをまとめて、 $h = \text{recserar}(a0 + \text{lagn}(e2, 1) * a1, \text{meanc}(e2), b1)$;ということになります。

ポイント $\text{recserar}(x, y0,)$ で、 x がスケーラーの場合、 $y_t = x + y_{t-1}$ が作られる
さらに、 x が列ベクトルの場合、 $y_t = x + y_{t-1} + y_{t-2} + \dots + y_{t-p}$

この場合、 x のところに $a0 + a_1 u_{t-1}^2$ がきていて、この部分が固定されてARプロセスがmeanc(e2)をもとに生成されます。ただし、これでは第1行目のシフトしたあとの0の入っているところもはいっているわけで、厳密ではありません。

さらに、0.95 confidence limits、すなわち95%の信頼区間の下限と上限を計算したものをパラメータの係数とともに表示させるのには、cmlprtを使うのではなくて、次のようにしてやります。

プログラム

```
new; cls;
library cml;
cmlset;
load data[100,2]=d:datafile14.txt;
data=data~ones(100,1);

_cml_Bounds={-1e256 1e256,
              -1e256 1e256,
              0.001 1e256,
              0      1,
              0      1};
_cml_c={0 0 0 -1 -1}; _cml_d= -0.9999;
_cml_ParNames = {"const","beta","a0","a1","b1"};
_cml_Algorithm=4;
b={0,0,1,1,1};
```

```

{x,f,g,cov,retcode}=cml(data,0,&ll,b);
cl=cmltlimits(x,cov);
call cmlclprt(x,f,g,cl,retcode);

proc ll(b,data);
  local y,x,e2,beta,a0,a1,b1,h;
  y=data[,1]; x=data[,3]~data[,2]; beta=b[1:2];
  e2=(y-x*beta)^2;
  a0=b[3]; a1=b[4]; b1=b[5];
  h=recserar(a0+lagn(e2,1)*a1, meanc(e2),b1);
  retp(-1/2*ln(2*pi)-1/2*ln(h)-1/2*e2./h);
endp;

```

画面表示

```

Mean log-likelihood      -0.482068
Number of cases          100

```

Parameters	Estimates	0.95 confidence limits		Gradient
		Lower Limit	Upper Limit	
const	0.4772	0.4024	0.5520	-0.0000
beta	1.9985	1.9727	2.0244	-0.0000
a0	0.0611	-0.0201	0.1423	0.0000
a1	0.2089	-0.1226	0.5404	0.0000
b1	0.4134	-0.0854	0.9123	0.0000

プログラムの中の太字の部分のように、**cml**は**cml**で別に計算します。そしてそのリターンを出すように書きます。すなわち、

```

{x,f,g,cov,retcode}=cml(data,0,&ll,b);

```

とします。5つあるリターンのうち、第1番目の**x**と第3番目の**cov**を使って、CMLのライブラリの中にある関数の**confidence limits**を求める**cmltlimits**を用いて、それぞれのパラメータの信頼区間を求めます。この**cmltlimits**では、グローバル変数**_cml_NumObs**でデータのそれぞれのシリーズの個数を設定しなければなりません。そして、**x**と**cov**をインプットして、**cl**として信頼区間がリターンされたものを、**cmlclprt**で表示します。その際、**cmlprt**とは違って、その第4インプットには**cl**がきていることに注意してください。このように**VarianceCovariance**行列ではなくて信頼区間が第4要素にくる場合には、通常の**cmlprt**で

はなくて、`cmlclprt`を用います。なお、`confidence limits`のデフォルトは、1-0.05の0.95、すなわち、95%信頼区間がデフォルトですが、これを、たとえば、

```
_cml_Alpha=0.01;
```

とすると、1-0.01の0.99、すなわち99%信頼区間が得られます。`_cml_Alpha`のグローバル変数の設定には、小数が使われることとともに、1からその数を引いたものが実際の信頼区間になるので注意が必要です。

ポイント `cl=cmltlimits(x,cov);` パラメータの列ベクトル `x`、`VarCov`行列が `cov`の時の各パラメータの95%信頼区間を求める。

その際には、`cmlprt`ではなく、`cmlclprt(x,f,g,cl,retcode)`で受けて表示する。

ただし、上のプログラムも一見うまくできているようですが、GAUSSのCMLデータハンドリングの仕様の視点からは、なおも不正確さが残ります。`_cml_NumObs`の設定で=99とすることによって100個の系列ではなくて99個で分析もできますが、これではウェイトをかけて初期の1データを殺していることにはなりません。そこで、各行にウェイトをかけて全99行のLog-Likelihoodを最大化します。正しくは、

```
__weight=(rows(data)/(rows(data)-1))*ones(rows(data),1); __weight[1]=zeros(1,1);
```

の行を加えて、各行のウェイトをデフォルトの1から変更します。100/99にしてやって、最初のデータのウェイトを0にします。(下線はそれぞれ2つ分です。)

画面表示

```
Mean log-likelihood      -0.485442
```

```
Number of cases      100
```

0.95 confidence limits				
Parameters	Estimates	Lower Limit	Upper Limit	Gradient

const	0.4754	0.4007	0.5500	-0.0000
beta	1.9979	1.9721	2.0237	-0.0000
a0	0.0605	-0.0194	0.1404	-0.0000
a1	0.2115	-0.1199	0.5429	-0.0000
b1	0.4152	-0.0756	0.9059	-0.0000

```
Number of iterations      45
```

```
Minutes to convergence      0.09783
```

結果は、前のものと比べて若干変わってきます。このように、各行のウェイトをデフォルトの1から、「全行数 / (全行数 - 1)」に若干増やしてやって、さらにオーバーライトする形

で、1行目だけのウエイトを0に再定義します。具体的には、下線2つにweightの__weightのグローバル変数を変更します。なお、n行分だけ1行目からウエイト0にするには、

```
__weight=(rows(data)/(rows(data)-n))*ones(rows(data),1); __weight[1:n]=zeros(n,1);
```

というふうに、分母で1を引くのではなくてnを引いてウエイトを増して、さらに、1行目からn行目までのウエイトにn行1列のすべてが0の要素を代入します。ここで、信頼区間を計算するグローバル変数_cml_NumObsの扱いですが、これは全100行ではなくて、計算上、実質上の行数の1減らした99とした数で計算した方が、より正確であるように思われます。係数には影響はありませんが、信頼区間の計算で値が変わってきます。非常に微妙な問題ですが、GARCH(p,q)のpまたはqのどちらか大きい方の一方の数が1ではなくて大きな数の場合に（上からカットする行数はpまたはqのどちらか大きい方の数）計算結果に無視できない影響を及ぼしてきます。通常は考えなくてもかまわないでしょう。

IGARCH(1,1)

さらに、GARCHの場合の最後の制約条件が等号条件、すなわち、

$$\sum_{i=1}^p b_i + \sum_{j=1}^q a_j = 1$$

の場合をIGARCHと言います。この場合のp=1,q=1のケースについて同じような計算をさせてみましょう。今度は、この場合、

$$b + a = 1$$

というかなりきつい制約になります。変更箇所は、不等号制約ではなくsqpSolveでやったように、等号制約の_cml_Aと_cml_Bの2つの行列を設定することになります。今、Dドライブにdatafile15.txtのファイルがあるものとします。

プログラム

```
new; cls;
library cml;
cmlset;
load data[100,2]=d:datafile15.txt;
data=data~ones(100,1);
```

```
_cml_Bounds={-1e256  1e256,
              -1e256  1e256,
```

```

0.001 1e256,
0      1,
0      1};
_cml_a={0 0 0 1 1}; _cml_b=1;
_cml_ParNames = {"const","beta","a0","a1","b1"};
_cml_Algorithm=4;
b={0,0,1,0.1,0.1};
__weight=(rows(data)/(rows(data)-1))*ones(rows(data),1); __weight[1]=zeros(1,1);
{x,f,g,cov,retcode}=cml(data,0,&ll,b);
_cml_NumObs=99;
cl=cmltlimits(x,cov);
call cmlclprt(x,f,g,cl,retcode);

```

```

proc ll(b,data);
  local y,x,e2,beta,a0,a1,b1,h;
  y=data[,1]; x=data[:,3]~data[:,2]; beta=b[1:2];
  e2=(y-x*beta)^2;
  a0=b[3]; a1=b[4]; b1=b[5];
  h=recserar(a0+lagn(e2,1)*a1, meanc(e2),b1);
  retp(-1/2*ln(2*pi)-1/2*ln(h)-1/2*e2./h);
endp;

```

画面表示

Mean log-likelihood -3.07455

Number of cases 99

0.95 confidence limits				
Parameters	Estimates	Lower Limit	Upper Limit	Gradient

const	0.5736	-0.1625	1.3098	0.0000
beta	1.8396	1.5693	2.1098	0.0000
a0	1.4381	-0.8037	3.6798	0.0000
a1	0.3668	0.0809	0.6526	-0.0574
b1	0.6332	0.3474	0.9191	-0.0574
Number of iterations	37			
Minutes to convergence	0.07600			

上のように、等号制約の行列 A および B を用いて計算すると、

$$_cml_A * b = _cml_B$$

すなわち、この場合、 $_cml_A$ は 1×5 、 b は 5×1 、 $_cml_B$ は 1×1 のスケーラーで、

$$\begin{bmatrix} 0 & 0 & 0 & 1 & 1 \end{bmatrix} \begin{bmatrix} const \\ beta \\ a0 \\ a1 \\ b1 \end{bmatrix} = 1$$

$1 \times a1 + 1 \times b1$ の制約を課すようになるように、 $_cml_a = \{0 \ 0 \ 0 \ 1 \ 1\}$; $_cml_b = 1$; というふうになっています。IGARCH への変更箇所は、これら 2 行だけです。

ポイント 等号制約 $_cml_A * b = _cml_B$
 不等号制約 $_cml_C * b > = _cml_D$

(例) b が 4 つのパラメータ $\alpha_1, \alpha_2, \alpha_3, \alpha_4$ の 4×1 の列ベクトルとすると、

$$\begin{aligned} \alpha_2 + \alpha_3 &= 1 \\ \alpha_3 + \alpha_4 &= 0 \end{aligned} \quad \text{の場合には}$$

$$_cml_A = \begin{bmatrix} 2 & 1 & 3 & 0 \\ 0 & 3 & 1 & -4 \end{bmatrix}, \quad _cml_B = \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$\begin{aligned} \alpha_1 + \alpha_2 + \alpha_3 &\leq 1 \\ 2\alpha_1 &\geq 1 \\ \alpha_1 - 2\alpha_2 &< 1 \end{aligned} \quad \text{の場合には}$$

$$_cml_C = \begin{bmatrix} -1 & -1 & -1 & -1 \\ 2 & 0 & 0 & 0 \\ 0 & -1 & 2 & 0 \end{bmatrix}, \quad _cml_D = \begin{bmatrix} -1 \\ 1 \\ -0.9999 \end{bmatrix}$$